Regular Paper

リーマンスティルチェス積分の形式化について
On the Formalizations of Riemann-Stieltjes Integral

師玉 康成 [1,*]
Yasunari Shidama[1,*]

1 信州大学工学部, 長野県長野市若里 4-17-1
1 Faculty of Engineering, Shinshu University. 4-17-1 Wakasato, Nagano, Japan

* shidama@cs.shinshu-u.ac.jp

Proof checked by Mizar Version: 8.1.08 and MML Version: 5.53.1335
Received: November 27, 2018. Accepted: April 11, 2019.

Abstract

In this paper, the definitions and basic properties of Riemann-Stieltjes integral are formalized in Mizar. In Section 2, we defined bounded variation function, total variation, and Riemann-Stieltjes integral of real valued function. In Section 3, we proved theorems about linearity of Riemann-Stieltjes integral. In the last section, the basic existence theorem of Riemann-Stieltjes integral is formalized. This theorem states that if u is a continuous function and v is a function of bounded variation in a closed interval of real line, u is Riemann-Stieltjes integrable with respect to v.

1 はじめに

リーマンスティルチェス積分 (Riemann–Stieltjes integral) の形式化について作業経過を報告する．この積分はリーマン積分の一般化である．現在，有界変動関数とリーマンスティルチェス積分の定義，連続関数の積分可能性定理，区間 $[a,b]$ 上の連続関数全体が作るバナハ空間 $C[a,b]$ の双対空間に関するリースの表現定理などの形式化が進捗している [1] [2] [3]．本稿では，連続関数の積分可能性までの概要を報告する．リーマンスティルチェス積分 (Riemann–Stieltjes integral) はリーマン積分の一般化である．本稿で報告する形式化も定義や定理は微積分の基礎的な文献 [4] [5] [6] などに沿ったものであり，形式化も既になされたリーマン積分の形式化 [7] [8] [9] を基に行っている．閉区間 $[a,b]$ の実数値関数 u,v についてリーマンスティルチェス積分は

$$\int_a^b u(t)dv(t) \tag{1}$$

と表されるがこれは，$[a,b]$ の分割 $t_1 = a < t_2 \cdots < t_n = b$ を用いたリーマンスティルチェス和

$$\sum_{1\le k<n} u(c_k)\{v(t_{k+1}) - v(t_k)\}, \quad c_k \in [t_k, t_{k+1}] \tag{2}$$

について $[a,b]$ の分割幅を無限に小さくする極限操作

$$\max\{|t_{k+1} - t_k|, 1 \le k < n\} \to 0 \quad (n \to \infty) \tag{3}$$

を行った (2) 式の極限値として定義される．

2 積分可能性と積分の定義の形式化

v が有界閉区間 $[a,b]$ 上の有界変動関数であるとは，正数 $0 < d$ が存在して $[a,b]$ の任意の分割 $t_1 = a < t_2 < t_3 \cdots < t_n = b$ について

$$\sum_{1\le k<n} |v(t_{k+1}) - v(t_k)| \le d \tag{4}$$

が成り立つことである．筆者らはこれを，有限列を用いて以下のように形式化した．

Listing 1. INTEGR22:def 1 def 2 def 3

```
Definition
  let A be Subset of REAL;
  let v be real-valued Function;
  func vol(A,v) -> Real equals
:: INTEGR22:def 1
  0 if A is empty otherwise
    v.(upper_bound A) - v.(lower_bound A);
end;
definition
  let A be non empty closed_interval Subset of REAL;
  let v be real-valued Function;
  let t be Division of A;
  mode var_volume of v,t -> FinSequence of REAL means
:: INTEGR22:def 2
    len it = len t
    & for k be Nat st k in dom t holds it.k = |. vol (divset(t,k),v) .|;
end;
definition
  let A be non empty closed_interval Subset of REAL;
  let v be Function of A,REAL;
  attr v is bounded_variation means
:: INTEGR22:def 3
  ex d be Real st
    0 < d
   &
    for t be Division of A, F be var_volume of v,t
 holds Sum(F) <= d;
end;
```

(注 1) 以後 INTEGR22, INTEGR23 の形式記述をいくつか提示するが，それらと本稿の記述の対応を判りやすくするため，一部の変数名を本稿の対応するものに書き替えするなどしている．

(注 2) 区間 $[a,b]$ の分割点列 $\{t_i\}_{1\le i\le n}$ について本稿では細分された閉区間は $[t_i, t_{i+1}]$ を用いるが，これに対応する INTEGER:22,23 では $divset(t,i) = [t_{i-1}, t_i](1 < i \le$

n), $divset(t,1) = [a,t_1]$ が使われている．$i = 1$ の場合の例外の記述を逐一書くと表現が煩雑になるため本稿では添え字 i を $i+1$ としている．同様に総和 $\sum_{1\le i<n} D_i$ なども INTEGER:22,23 では $\sum_{1<i\le n} D_{i-1}$ が対応する．

ここで vol(divset(t,k),v) と表現されているものは，既に MML に収録されているリーマン積分の形式化定義 [7] の流用で $[a,b]$ の任意の分割 $t_1 = a < t_2 < t_3 \cdots < t_n = b$ について

$$v(t_{k+1}) - v(t_k) \tag{5}$$

を表している．また，これを用いた数列

$$\{F_k\}_{1\le k<n}, \quad F_k = |v(t_{k+1}) - v(t_k)| \tag{6}$$

を導入している．$[a,b]$ の分割 $t_1 = a < t_2 < t_3 \cdots < t_n = b$ 全てについて

$$\sum_{1\le k<n} |v(t_{k+1}) - v(t_k)| \tag{7}$$

の上限は全変動と呼ばれるが，これを以下のように形式化した．

Listing 2. INTEGR22:def 4

```
Definition
  let A be non empty closed_interval Subset of REAL;
  let v be Function of A,REAL;
  assume
 v is bounded_variation;
  func total_vd(v) -> Real means
:: INTEGR22:def 4
  ex VD be non empty Subset of REAL st
    VD is bounded_above
   &
    VD = { r where r is Real:
      ex t be Division of A, F be var_volume of v,t st
 r = Sum(F) }
  & it = upper_bound VD;
end;
```

(2) 式のリーマンスティルチェス和の表現には以下のように有限列を用いた形式化を行った．

Listing 3. INTEGR22:def 5

```
Definition
  let A be non empty closed_interval Subset of REAL;
  let v be Function of A,REAL;
  let u be PartFunc of REAL,REAL;
  assume
 v is bounded_variation & dom u = A;
  let t be Division of A;
  mode middle_volume of v,u,t -> FinSequence of REAL means
:: INTEGR22:def 5
  len it = len t
  & for k be Nat st k in dom t holds ex r be Real st
  r in rng (u|divset(t,k))
  & it.k = r*(vol (divset(t,k),v));
end;
```

関数 u, v について分割された各小区間 $[t_k, t_{k+1}]$ の元 c_k は一意には定まらない．各小区間 $[t_k, t_{k+1}]$ の u による像の中から適当な元 r を選び（これは $c_k \in [t_k, t_{k+1}]$ を一つ選び $r = u(c_k)$ とすることと同じであり），それと $v(t_{k+1}) - v(t_k)$ の積を値にもつ実数の有限列を構成し，これを middle_volume_Sequence of v,u,T という mode で定義した．divset(t,k) は分割された小区間 $[t_k, t_{k+1}]$ を表している．

この実数の有限列 middle_volume_Sequence of v,u,T の和がリーマンスティルチェス和である．区間 $[a, b]$ の分割数を増して分割幅を無限に小さくしたときの (2) 式の極限を求める手続きを形式化するには分割の無限列を用いる．以下のように無限列を用いた形式化を行った．

Listing 4. INTEGR22:def 6

```
Definition
  let A be non empty closed_interval Subset of REAL;
  let v be Function of A,REAL;
  let u be PartFunc of REAL,REAL;
::: assume v is bounded_variation & dom u = A;
  let T be DivSequence of A;
  mode middle_volume_Sequence of v,u,T -> sequence of (REAL)* means
:: INTEGR22:def 6
    for k be Element of NAT holds it.k is middle_volume of v,u,T.k;
end;
```

この middle_volume_Sequence of v,u,T は区間 $[a, b]$ の分割の無限列

$$T = \{\{T_i^k\}_{1 \le i \le n_k}\}_{k \in N}, \quad T_1^k = a < T_2^k < T_3^k \cdots < T_{n_k}^k = b \tag{8}$$

が与えられたとき，各 k について，i 番目の値が

$$u(c_i^k)\{v(T_{i+1}^k) - v(T_i^k)\}, \quad c_i^k \in [T_i^k, T_{i+1}^k] \tag{9}$$

である実数数列を表している．これも c_i^k の選択が一意に決まらないため mode として定義している．

S を middle_volume_Sequence of v,u,T とするとき，各 k について実数数列 S_k の和を値にもつ実数数列 middle_sum(S) を以下のように定義した．

Listing 5. INTEGR22:def 7

```
definition
  let A be non empty closed_interval Subset of REAL;
  let v be Function of A,REAL;
  let u be PartFunc of REAL,REAL;
::: assume v is bounded_variation & dom u = A;
  let T be DivSequence of A;
  let S be middle_volume_Sequence of v,u,T;
  func middle_sum(S) -> Real_Sequence means
:: INTEGR22:def 7
  for k be Nat holds it.k = Sum(S.k);
end;
```

この実数数列 middle_sum(S) の各 k での値は，分割 T^k についての u, v のリーマンスティルチェス和

$$\sum_{1\le i<n_k} u(c_i^k)\{v(T_{i+1}^k)-v(T_i^k)\},\quad c_i^k\in[T_i^k,T_{i+1}^k] \tag{10}$$

である．

以上のもとに実数値関数 u の関数 v についてのリーマンスティルチェス積分の定義を以下のように形式化した．u が v についてリーマンスティルチェス積分可能であるとは，実数 I が存在し，区間 $[a,b]$ の任意の分割の無限列

$$T=\{\{T_i^k\}_{1\le i\le n_k}\}_{k\in N},\quad T_1^k=a<T_2^k<T_3^k\cdots<T_{n_k}^k=b$$

で lim delta(T) = 0，すなわち

$$\max\{|T_{i+1}^k-T_i^k|, 1\le i<n_k\}\to 0\quad(k\to\infty) \tag{11}$$

を満たすものについて，各 k について分割 T^k による小区間の代表元の列 $\{c_i^k\}_{1\le i<n_k}$ を任意に選んで作られるリーマンスティルチェス和の実数数列 (middle_volume_Sequence of v,u,T) が収束し，いずれも I をその収束値にもつことであり，これを述語 u is_RiemannStieltjes_integrable_with v で表現する．さらにこの条件が成り立つとき一意に存在する I を u の v についてのリーマンスティルチェス積分

$$\int_a^b u(t)dv(t) \tag{12}$$

と定義しこれを u, v を引数のもつ integral(u,v) という Functor で表す．

Listing 6. INTEGR22:def 8 def 9

```
Definition
  let A be non empty closed_interval Subset of REAL;
  let v be Function of A,REAL;
  let u be PartFunc of REAL,REAL;
::: assume v is bounded_variation & dom u = A;
  pred u is_RiemannStieltjes_integrable_with v means
:: INTEGR22:def 8
  ex I be Real st for T being DivSequence of A,
      S be middle_volume_Sequence of v,u,T st
      delta(T) is convergent & lim delta(T) = 0 holds
      middle_sum(S) is convergent & lim (middle_sum(S)) = I;
end;

definition
  let A be non empty closed_interval Subset of REAL;
  let v be Function of A,REAL;
  let u be PartFunc of REAL,REAL;
  assume v is bounded_variation & dom u = A &
u is_RiemannStieltjes_integrable_with v;
  func integral(u,v) -> Real means
:: INTEGR22:def 9
  for T being DivSequence of A,
S be middle_volume_Sequence of v,u,T st
  delta(T) is convergent & lim delta(T) = 0 holds
  middle_sum(S) is convergent & lim (middle_sum(S)) = it;
end;
```

3 積分の線形性

この定義をもとに，実数値関数 u の関数 v についてのリーマンスティルチェス積分に関して，関数の加法，スカラ倍に関しての積分の線形性について以下の命題の形式化と証明を行った．例えば実数値関数 u, w が v についてリーマンスティルチェス積分可能とするとき，前節の

$$T = \{\{T_i^k\}_{1 \le i \le n_k}\}_{k \in N}, \quad T_1^k = a < T_2^k < T_3^k \cdots < T_{n_k}^k = b$$

で lim delta(T) = 0 すなわち (11) 式

$$\max\{|T_{i+1}^k - T_i^k|, 1 \le i < n_k\} \to 0 \quad (k \to \infty)$$

を満たすものについて，関数の和 $u + w$ に関して各 k 番目の項がリーマンスティルチェス和

$$\sum_{1 \le i < n_k} (u + \mathrm{w})(c_i^k)\{v(T_{i+1}^k) - v(T_i^k)\}, \quad c_i^k \in [T_i^k, T_{i+1}^k] \tag{13}$$

になる実数列を作れば，それぞれ各 k 番目の項が

$$\sum_{1 \le i < n_k} u(c_i^k)\{v(T_{i+1}^k) - v(T_i^k)\}, \sum_{1 \le i < n_k} w(c_i^k)\{v(T_{i+1}^k) - v(T_i^k)\}, \quad c_i^k \in [T_i^k, T_{i+1}^k] \tag{14}$$

である実数列が作れる．(11) 式，すなわち lim delta(T) = 0 が成り立っているため，また u, w が v についてリーマンスティルチェス積分可能であることから，(14) の 2 つの級数は $k \to \infty$ とするとき収束する．

$$\begin{aligned}\sum_{1 \le i < n_k} (u + \mathrm{w})(c_i^k)\{v(T_{i+1}^k) - v(T_i^k)\} &= \sum_{1 \le i < n_k} u(c_i^k)\{v(T_{i+1}^k) - v(T_i^k)\} \\ &+ \sum_{1 \le i < n_k} w(c_i^k)\{v(T_{i+1}^k) - v(T_i^k)\} \end{aligned} \tag{15}$$

が成り立っているから，(14) の収束性により (15) 式の両辺の級数は $k \to \infty$ とするとき収束し，その極限は

$$\int_a^b (u + w)(t)dv(t) = \int_a^b u(t)dv(t) + \int_a^b w(t)dv(t)$$

となる（本稿の形式化では integral(u+w,v) = integral(u,v) + integral(w,v) と表される）．以下同様の議論により以下の命題群を形式化した．

Listing 7. INTEGR22:7∼17

```
begin :: Linearity of Riemann−Stieltjes Integral
theorem :: INTEGR22:7
  for A be non empty closed_interval Subset of REAL,
r be Real,
v be Function of A,REAL,
u,w be PartFunc of REAL,REAL st
v is bounded_variation & dom u = A & dom w = A &
w = r(#)u & u is_RiemannStieltjes_integrable_with v holds
  w is_RiemannStieltjes_integrable_with v &
```

```
  integral(w,v) = r * integral(u,v);

theorem :: INTEGR22:9
  for A be non empty closed_interval Subset of REAL,
v be Function of A,REAL,
u,s,w be PartFunc of REAL,REAL st
v is bounded_variation & dom u = A & dom v = A & dom s = A &
w = u + s & u is_RiemannStieltjes_integrable_with v &
v is_RiemannStieltjes_integrable_with v holds
  s is_RiemannStieltjes_integrable_with v &
  integral(w,v) = integral(u,v) + integral(s,v);

theorem :: INTEGR22:12
  for A be non empty closed_interval Subset of REAL,
r be Real, v,v1 be Function of A,REAL,
u be PartFunc of REAL,REAL st
v is bounded_variation & v1 is bounded_variation & dom u = A &
v = r(#)v1 & u is_RiemannStieltjes_integrable_with v1 holds
  u is_RiemannStieltjes_integrable_with v &
  integral(u,v) = r * integral(u,v1);

theorem :: INTEGR22:15
  for A be non empty closed_interval Subset of REAL,
v,v1,v2 be Function of A,REAL,
u be PartFunc of REAL,REAL st
v is bounded_variation & v1 is bounded_variation &
v2 is bounded_variation & dom u = A &
v = v1 + v2 & u is_RiemannStieltjes_integrable_with v1 &
u is_RiemannStieltjes_integrable_with v2
 holds
  u is_RiemannStieltjes_integrable_with v &
  integral(u,v) = integral(u,v1) + integral(u,v2);
```

なお，上記記述で v が有界変動であるという条件 v is bounded_variation は不要であり，現在，MML への修正削除版を準備中である．近々削除される予定である．

4 連続関数の積分可能性

閉区間上の連続な実数値関数 u と有界変動関数 v についてはリーマンスティルチェス積分可能である．これは以下のように形式化した．

Listing 8. INTEGR23:21

```
theorem :: INTEGR23:21
  for A be non empty closed_interval Subset of REAL,
    v be Function of A,REAL,
    u be continuous PartFunc of REAL,REAL st
    v is bounded_variation & dom u = A
   holds u is_RiemannStieltjes_integrable_with v;
```

この証明には，実数 I が存在し，区間 $[a,b]$ の任意の分割の無限列

$$T = \{\{T_i^k\}_{1\leq i\leq n_k}\}_{k\in N},\quad T_1^k = a < T_2^k < T_3^k \cdots < T_{n_k}^k = b$$

で lim delta(T) = 0 すなわち

$$\max\{|T_{i+1}^k - T_i^k|, 1\leq i < n_k\} \to 0 \quad (k\to\infty)$$

を満たすものについて，各 k について分割 T^k による小区間の代表元の列 $\{c_i^k\}_{1\le i<n_k}$ を任意に選んで作られるリーマンスティルチェス和の実数数列 middle_volume_Sequence of v,u,T

$$\{S_k\}_{k\in N},\quad S_k=\sum_{1\le i<n_k} u(c_i^k)\{v(T_{i+1}^k)-v(T_i^k)\},\quad c_i^k\in[T_i^k,T_{i+1}^k] \tag{16}$$

が収束し，その極限が I と一致することを示す必要がある．u は有界閉区間 $[a,b]$ 上の連続関数だから一様連続である．

v は有界変動関数であるから任意の k について

$$\sum_{1\le i<n_k}|v(T_{i+1}^k)-v(T_i^k)|\le total\_vd(v) \tag{17}$$

が成り立っている．$0<\epsilon$ を任意の正の実数とするとき，関数 u の $[a,b]$ 上の一様連続性から，ある正数 $0<\delta$ が存在し，任意の実数 $t_1,t_2\in[a,b]$ について

$$|t_1-t_2|<\delta\quad ならば\quad |u(t_1)-u(t_2)|<\epsilon/(total\_vd(v)+1) \tag{18}$$

が成り立っている．一方，lim delta(T) = 0 が成り立っているから自然数 m が存在し，$m\le k$ である任意の自然数 k について

$$\max\{|T_{i+1}^k-T_i^k|,1\le i<n_k\}<\delta \tag{19}$$

が成り立っている．

ここで $m\le k$ である任意の自然数 k について (18), (19) 式から

$$|\ S_k-S_m|<\epsilon$$

が成り立っていることを示せば，リーマンスティルチェス和の列 $\{S_k\}_{k\in N}$ がコーシー列になることから，実数空間の完備性により $\{S_k\}_{k\in N}$ は収束することが証明できる．

しかしながら

$$\big|S_k-S_m\big|\quad=\Big|\sum_{1\le i<n_k}u(c_i^k)\{v(T_{i+1}^k)-v(T_i^k)\}-\sum_{1\le j<n_m}u(c_j^m)\{v(T_{j+1}^m)-v(T_j^m)\}\Big|$$

の評価はそれ程容易ではない．区間分割数 n_k と n_m が等しいとは限らないし，$i=j$ であっても，それぞれの区間 $[T_i^k,T_{i+1}^k],[T_j^m,T_{j+1}^m]$ やそれらの代表元 c_i^k,c_j^m が同じであるとは限らない．第 2 節で述べた表記方法を用いると

$$\begin{aligned}divset(T^k,i)&=[T_i^k,T_{i+1}^k],\quad vol(divset(T^k,i),v)=v(T_{i+1}^k)-v(T_i^k)\\ divset(T^m,j)&=[T_j^m,T_{j+1}^m],\quad vol(divset(T^m,j),v)=v(T_{j+1}^m)-v(T_j^m)\end{aligned}$$

であり，S_k,S_m はそれぞれ

$$S_k=\sum_{1\le i<n_k}u(c_i^k)vol(divset(T^k,i),v) \tag{20}$$

$$S_m=\sum_{1\le j<n_m}u(c_j^m)vol(divset(T^m,j),v) \tag{21}$$

と書ける．ここで

$$[a,b] = \bigcup_{1\le j<n_m} [T^m_j, T^m_{j+1}] = \bigcup_{1\le j<n_m} divset(T^m, j)$$

から

$$\begin{aligned} divset(T^k,i) = divset(T^k,i)\cap[a,b] &= divset(T^k,i)\cap\{\bigcup_{1\le j<n_m} divset(T^m,j)\} \\ &= \bigcup_{1\le j<n_m}\{divset(T^k,i)\cap divset(T^m,j)\} \quad (22)\end{aligned}$$

であり

$$vol(divset(T^k,i),v) = \sum_{1\le j<n_m} vol(divset(T^k,i)\cap divset(T^m,j),v) \quad (23)$$

全く同様にして

$$vol(divset(T^m,j),v) = \sum_{i\le i<n_k} vol(divset(T^k,i)\cap divset(T^m,j),v) \quad (24)$$

を得る．

ここで区間 $[T^k_i, T^k_{i+1}]$ と $[T^m_j, T^m_{j+1}]$ が共通部分をもたないか，あるいは端点一点のみを共有する場合には

$$vol(divset(T^k,i)\cap divset(T^m,j),v) = 0$$

であり，$[T^k_i, T^k_{i+1}]$ と $[T^m_j, T^m_{j+1}]$ が共通部分を持つ場合は (18), (19) 式から

$$|u(c^k_i) - u(c^m_j)| < \epsilon/(total\_vd(v)+1) \quad (25)$$

が従う．以下の補題 INTEGR22:2

Listing 9. INTEGR22:2

```
theorem :: INTEGR22:2
  for n,m be Nat, a be Function of [:Seg n,Seg m:],REAL
  for p,q be FinSequence of REAL st
    ( dom p = Seg n
    & for i be Nat st i in dom p holds
        ex r be FinSequence of REAL st
          (dom r = Seg m & p.i = Sum r
         & for j be Nat st j in dom r holds r.j=a.(i,j) ) )
  & ( dom q = Seg m
    & for j be Nat st j in dom q holds
        ex s be FinSequence of REAL st
          (dom s = Seg n & q.j = Sum s
         & for i be Nat st i in dom s holds s.i=a.(i,j) ) )
  holds Sum p = Sum q;
```

を用いれば総和 $\sum_{1\le j<n_m}$ と $\sum_{1\le i<n_k}$ の順序を変えることができ，(20), (21), (23), (24) 式から

$$S_k - S_m = \sum_{1\le i<n_k}\sum_{1\le j<n_m} (u(c^k_i) - u(c^m_j))vol(divset(T^k,i)\cap divset(T^m,j),v) \quad (26)$$

である．また，

$$\sum_{1\leq i<n_k}\sum_{1\leq j<n_m}|vol(divset(T^k,i)\cap divset(T^m,j),v)|\leq total\_vd(v)$$

である．この評価式と (26), (25) 式から目的の

$$|S_k-S_m|\leq \quad \epsilon/(total\_vd(v)+1)\sum_{1\leq i<n_k}\sum_{1\leq j<n_m}|vol(divset(T^k,i)\cap divset(T^m,j),v)|<\epsilon$$

を得る．これらは下記のように補題として書かれている．

Listing 10. INTEGR23:17 INTEGR23:18

```
theorem :: INTEGR23:17
  for A be non empty closed_interval Subset of REAL,
    rho be Function of A,REAL,
    T,S be Division of A
  st rho is bounded_variation holds
   ex ST be FinSequence of REAL
   st len ST = len S
    & Sum ST <= total_vd(rho)
    & for j be Nat st j in dom S holds
      ex p be FinSequence of REAL
      st ST.j = Sum p
       & len p = len T
       & for i be Nat st i in dom T holds
         p.i = |. vol(divset(T,i) /\ divset(S,j),rho) .|;

theorem :: INTEGR23:18
  for A be non empty closed_interval Subset of REAL,
    rho be Function of A,REAL,
    u be PartFunc of REAL,REAL
  st rho is bounded_variation
   & dom u = A
   & u|A is uniformly_continuous
  holds
    for T being DivSequence of A, S be middle_volume_Sequence of rho,u,T
     st delta T is convergent & lim delta T = 0
    holds middle_sum(S) is convergent;
```

証明の詳細は `http://www.mizar.org/version/current/html/integr23.html` に公開されている．証明記述の殆どが，有限列と無限列の構成と，それらに関する手続きの正当性を示すための数学的帰納法によって占められている．

次に，この極限値が lim delta(T) = 0 を満たす区間 $[a,b]$ の分割と分割された各小区間の代表元の選び方に依存しないで，一意に決まることを示す．それぞれ lim delta(P) = 0, lim delta(Q) = 0 を満たす区間 $[a,b]$ の分割の無限列 $\{P^k\}_{k\in N},\{Q^k\}_{k\in N}$

$$P^k_1=a<P^k_2<P^k_3\cdots<P^k_{n_k}=b, Q^k_1=a<Q^k_2<Q^k_3\cdots<Q^k_{\mathrm{m}_k}=b \qquad (27)$$

と分割される小区間の代表元の列

$$\begin{aligned}&\{\{d^k_i\}_{1\leq i<n_k}\}_{k\in N},\quad d^k_i\in[P^k_i,P^k_{i+1}]\quad(1\leq i<n_k)\\ &\{\{e^k_i\}_{1\leq i<m_k}\}_{k\in N},\quad e^k_i\in[Q^k_i,Q^k_{i+1}]\quad(1\leq i<m_k)\end{aligned} \qquad (28)$$

が任意に与えられたとする．これによるリーマンスティルチェス和の列

$$U_k = \sum_{1 \le i < n_k} u(d_i^k)\{v(P_{i+1}^k) - v(P_i^k)\}, \quad W_k = \sum_{1 \le i < m_k} u(e_i^k)\{v(Q_{i+1}^k) - v(Q_i^k)\} \tag{29}$$

は前述のようにコーシー列になることから収束する．証明するべきことは

$$\lim_{k \to \infty} U_k = \lim_{k \to \infty} W_k$$

である．上の分割の無限列から分割の無限列 $\{T^k\}_{k \in N}$ を

$$T^{2k} = P^k, \quad T^{2k+1} = Q^k \tag{30}$$

として構成すると，lim delta(T) = 0 を満たしている．さらにこの $\{T^k\}_{k \in N}$ により分割された各小区間の代表元の列 $\{\{c_i^k\}_{1 \le i < l_k}\}_{k \in N}$ を以下のように作る

$$\begin{aligned} c_i^{2k} &= d_k^i \quad (1 \le i < l_{2k}) \quad l_{2k} = n_k \\ c_i^{2k+1} &= e_k^i \quad (1 \le i < l_{2k+1}) \quad l_{2k+1} = m_k \end{aligned} \tag{31}$$

これらから作られるリーマンスティルチェス和の列

$$\{S_k\}_{k \in N}, \quad S_k = \sum_{1 \le i < l_k} u(c_i^k)\{v(T_{i+1}^k) - v(T_i^k)\} \tag{32}$$

を作ると，前述のとおり $\{S_k\}_{k \in N}$ はコーシー列であるため収束する．さらに分割の無限列 $\{P^k\}_{k \in N}, \{Q^k\}_{k \in N}$ と分割される小区間の代表元の列 $\{\{d_i^k\}_{1 \le i < n_k}\}_{k \in N}, \{\{e_i^k\}_{1 \le i < m_k}\}_{k \in N}$ から作られる (29) 式のリーマンスティルチェス和の列は

$$S_{2k} = U_k, \quad S_{2k+1} = W_k$$

となるので $\{S_k\}_{k \in N}$ の部分列である．収束列 $\{S_k\}_{k \in N}$ の部分列 $\{U_k\}_{k \in N}, \{W_k\}_{k \in N}$ の極限はいずれも $\{S_k\}_{k \in N}$ の極限と一致する．

以下はこの証明に使用された補題の形式化の一部である．

Listing 11. INTEGR23:21 INTEGR23:22 (excerpt)

```
theorem Th21:
  for A be non empty closed_interval Subset of REAL,
    v be Function of A,REAL,
    u be PartFunc of REAL,REAL,
    T0,T,T1 be DivSequence of A,
    S0 be middle_volume_Sequence of v,u,T0,
    S be middle_volume_Sequence of v,u,T
  st for i be Nat holds T1.(2*i) = T0.i & T1.(2*i+1) = T.i
  ex S1 be middle_volume_Sequence of v,u,T1
  st for i be Nat holds S1.(2*i) = S0.i & S1.(2*i+1) = S.i

theorem Th22:
  for Sq0,Sq,Sq1 be Real_Sequence
  st Sq1 is convergent
   & for i be Nat holds Sq1.(2*i) = Sq0.i & Sq1.(2*i+1) = Sq.i holds
    Sq0 is convergent & lim Sq0 = lim Sq1
  & Sq is convergent & lim Sq = lim Sq1
```

以上が形式化証明の概要であるが，区間の無限分割に関する議論で，分割，結合，再分割の操作などで読者の直観に訴えるような方法は厳密な形式化証明では許容されない．有限回, 無限回の操作に対応した，有限列，無限列 [10] [11] の構成が必要になる．繰り返し操作の正当性は数学的帰納法 [12] によっている．

5　終わりに

本稿ではリーマンスティルチェス積分 (Riemann–Stieltjes integral) の形式化について作業経過を報告した. 有界変動関数とリーマンスティルチェス積分の定義，連続関数の積分可能性定理の形式化の状況を解説した．区間 $[a, b]$ 上の連続関数全体が作るバナハ空間 $C[a, b]$ の双対空間に関するリースの表現定理などの形式化が進捗しているが，これについては次回に報告する．

参考文献

[1] Narita K, Nakasho K, Shidama Y. Riemann-Stieltjes Integral. Formalized Mathematics. 2016;24(**3**):199–204.

[2] Nakasho K, Narita K, Shidama Y. The Basic Existence Theorem of Riemann-Stieltjes Integral. Formalized Mathematics. 2016;24(**4**):253–259.

[3] Narita K, Nakasho K, Shidama Y. F. Riesz Theorem. Formalized Mathematics. 2017;25(**3**):179–184.

[4] Gupta SL, Rani N. Fundamental Real Analysis. Vikas Pub.; 1986.

[5] Kestelman H. Methods in classical and functional analysis. Addison-Wesley Publishing Co.; 1960.

[6] Stroock DW. A Concise Introduction to the Theory of Integration. Springer Science and Business Media; 1999.

[7] Endou N, Korniłowicz A. The Definition of the Riemann Definite Integral and some Related Lemmas. Formalized Mathematics. 1999;8(**1**):93–102. Available from: `http://fm.mizar.org/1999-8/pdf8-1/integra1.pdf`.

[8] Endou N, Wasaki K, Shidama Y. Darboux's Theorem. Formalized Mathematics. 2001;9(**1**):197–200. Available from: `http://fm.mizar.org/2001-9/pdf9-1/integra3.pdf`.

[9] Miyajima K, Shidama Y. Riemann Integral of Functions from $\mathbb{R}$ into $\mathcal{R}^\backslash$. Formalized Mathematics. 2009;17(**2**):179–185.

[10] Bancerek G, Hryniewiecki K. Segments of Natural Numbers and Finite Sequences. Formalized Mathematics. 1990;1(**1**):107–114. Available from: `http://fm.mizar.org/1990-1/pdf1-1/finseq_1.pdf`.

[11] Kotowicz J. Real Sequences and Basic Operations on Them. Formalized Mathematics. 1990;1(**2**):269–272. Available from: `http://fm.mizar.org/1990-1/pdf1-2/seq_1.pdf`.

[12] Bancerek G. The Fundamental Properties of Natural Numbers. Formalized Mathematics. 1990;1(**1**):41–46. Available from: `http://fm.mizar.org/1990-1/pdf1-1/nat_1.pdf`.

Mizar article information

Works in Progress

INTEGR22 Riemann-Stieltjes Integral
by Keiko Narita, Kazuhisa Nakasho, and Yasunari Shidama

INTEGR23 The Basic Existence Theorem of Riemann-Stieltjes Integral
by Kazuhisa Nakasho, Keiko Narita, and Yasunari Shidama

DUALSP05 F. Riesz Theorem
by Keiko Narita, Kazuhisa Nakasho, and Yasunari Shidama